

gnuplot



© Predrag Pejović,



gnuplot . . .

- ▶ dugovečan program (1986) za vizuelizaciju podataka, stalno dograđivan
- ▶ ugrađen u GNU Octave, Maxima, R, . . .
- ▶ crta podatke (iz fajla) i funkcije (analitički zadate)
- ▶ 2D i 3D, eksplicitno i parametarski
- ▶ Dekartov i polarni koordinatni sistem
- ▶ komandna linija, danas za većinu korisnika problem
- ▶ veoma kvalitetan izlaz
- ▶ razvija se: nove verzije, knjige . . .
- ▶ mnogima težak, mada je sve to ista stvar . . .
- ▶ vole ga statističari i zainteresovani za advanced data analysis

... gnuplot ...

- ▶ glavni izvor: <http://www.gnuplot.info/>
- ▶ *sada* aktuelna verzija 5.2.2
- ▶ za win sada instalacija, ...
- ▶ uputstvo
http://www.gnuplot.info/docs_5.2/Gnuplot_5.2.pdf,
271 strana
- ▶ pomoćni izvor: <http://www.gnuplotting.org/>
- ▶ proprietary:
 - ▶ Gnuplot in Action
 - ▶ gnuplot Cookbook
 - ▶ gnuplot 5
- ▶ očigledno se pišu knjige na gnuplot temu ...


... gnuplot

- ▶ program stabilan, sa mnogo korisnika, ima smisla učiti ga
- ▶ veoma puno resursa dostupno
- ▶ sjajna zbirka primera <http://gnuplot.sourceforge.net/demo/>
- ▶ odaberete sta vam treba, preradite, brzo i efikasno, moj slučaj, Madrid ...
- ▶ **alternative:**
 - ▶ Origin, US\$ 1095 (std), US\$ 1800 (pro), US\$ 850 (academic pro), US\$ 69/yr (student)
 - ▶ SigmaPlot, sa njihovog sajta: Academic: \$599, Commercial: \$899, Government: \$799
- ▶ imate neke primedbe na komandnu liniju?


gnuplot, alternative ...

- ▶ jedna od GUI free alternativa **SciDAVis** (Python)
- ▶ jos poznati **LabPlot**, **Grace**, **Veusz** (Python)
- ▶ **PyXPlot**, vrlo slicna sintaksa (Python)
- ▶ ...
- ▶ treba li vam ista od toga?
- ▶ ja mislim: **ne**
- ▶ lično iskustvo: što standardnije, to bolje
- ▶ lično iskustvo: što veća baza korisnika, to bolje
- ▶ slike se recikliraju ...


gnuplot, za šta je dobar?




gnuplot, za šta je dobar?



gnuplot, za šta je dobar?



gnuplot, za šta je dobar?



gnuplot, za šta je dobar?



gnuplot, za šta je dobar?



gnuplot, terminal -> terminal

- ▶ idemo na komandnu liniju, crveno su gnuplot komande koje kucate
- ▶ `gnuplot`
- ▶ `help`
- ▶ `plot sin(x)/x`
- ▶ `print pi`
- ▶ `print 5*pi`
- ▶ `print 2**3`
- ▶ ima i promenljive
- ▶ `a=pi/4`
- ▶ `print sin(a)**2`

gnuplot, xrange, yrange

- ▶ `set xrange [-5*pi:5*pi]`
- ▶ `show xrange`
- ▶ `replot`
- ▶ `set yrange [-1.1:1.1]`
- ▶ `show yrange`
- ▶ `replot`
- ▶ `reset`
- ▶ `replot`

gnuplot, samples

- ▶ `show samples`
- ▶ `set samples 10`
- ▶ `plot sin(x), cos(x)`
- ▶ `show samples`
- ▶ `set samples 360*5`
- ▶ `replot`
- ▶ `show samples`
- ▶ `help set samples`

gnuplot, more functions, key

- ▶ `plot sin(x)/x, -sin(x)/x`
- ▶ `unset key`
- ▶ `replot`
- ▶ `set key left top`
- ▶ `replot`
- ▶ `set key right bottom box`
- ▶ `replot`
- ▶ `help set key`
- ▶ `set key tmargin left` (ostaje box!)
- ▶ `replot`
- ▶ `set key rmargin center nobox`
- ▶ `replot`
- ▶ `set key above box; replot`
- ▶ `set key below nobox; replot`
- ▶ ima još opcija, `help`, `books`, ...

gnuplot, tics

- ▶ `plot [-2*pi:2*pi] sin(x), sin(x)**2`
- ▶ `set xtics (-2*pi, -pi, 0, pi, 2*pi); replot`
- ▶ `set xtics nomirror; replot`
- ▶ `set grid; replot`
- ▶ `set xtics (-2*pi, -3*pi/2, -pi, -pi/2, 0, pi, 2*pi); replot`
- ▶ `unset grid; replot` (nomirror ostaje!)
- ▶ `set xtics ("-pi" -pi, "0" 0, "pi" pi); replot`
- ▶ `reset; replot`
- ▶ `set xtics pi`
- ▶ `set mxtics 2; replot`
- ▶ `set grid mxtics; replot`
- ▶ `set grid xtics mxtics ytics; replot`

gnuplot, lines and points

- ▶ `reset; plot sin(x), cos(x)`
- ▶ `plot sin(x) with lines linetype 1 linewidth 1, cos(x) with lines linetype 1 linewidth 2`
- ▶ `plot sin(x) with lines linetype 3 linewidth 2, cos(x) with lines linetype 4 linewidth 1`
- ▶ `plot sin(x) with points`
- ▶ `plot sin(x) with linespoints`
- ▶ `plot sin(x) with dots`
- ▶ `plot sin(x) with points, cos(x) with points`
- ▶ `set samples 10`
- ▶ `set yrange [0:6]`
- ▶ `plot 1 with points pointtype 1, \`
`2 with points pointtype 2, \`
`3 with points pointtype 3, \`
`4 with points pointtype 4`

gnuplot, steps and bars

- ▶ `reset; plot sin(x) with impulses`
- ▶ `plot sin(x) with steps`
- ▶ `plot sin(x) with steps, sin(x) with fsteps, sin(x) with histeps`
- ▶ `plot sin(x) with boxes`
- ▶ `set style fill solid 1; replot`
- ▶ `set boxwidth 0.5 relative; replot`

gnuplot, labels

- ▶ `plot sin(x)`
- ▶ `set xlabel "x"`
- ▶ `set ylabel "y"`
- ▶ `set title "y=sin(x)"`
- ▶ `replot`
- ▶ `set label "neki tekst" at 5,0.5; replot`
- ▶ `plot sin(x) title "sin", cos(x) title "cos"`
- ▶ `set key box; replot`

gnuplot, terminals and output 1

- ▶ `show terminal`
- ▶ `help terminal`
- ▶ `help terminal postscript`
- ▶ `set terminal postscript eps`
- ▶ `show output`
- ▶ `set output "slika1.ps"`
- ▶ `plot sin(x), cos(x)`
- ▶ pogledajte šta ste dobili kao `slika1.ps`
- ▶ `set output "slika2.ps"`
- ▶ `set terminal postscript eps color`
- ▶ `plot sin(x), cos(x)`
- ▶ pogledajte šta ste dobili kao `slika2.ps`

gnuplot, terminals and output 2

- ▶ `set output`
- ▶ `show output`
- ▶ `show terminal`
- ▶ menjate terminal koji imate, qt ili wxt, ja pišem za wxt, važi i na dalje
- ▶ `set terminal wxt 0 persist`
- ▶ `plot sin(x)`
- ▶ `set terminal wxt 1 persist`
- ▶ `plot cos(x)`
- ▶ `set terminal wxt 0`
- ▶ `show terminal`
- ▶ `reset`

gnuplot, da napravimo file

- ▶ `octave`
- ▶ `deg=0:360*4;`
- ▶ `wt=deg*pi/180;`
- ▶ `f=exp(-wt/2/pi*0.5).*sin(wt);`
- ▶ `plot(deg,f)`
- ▶ `fp=exp(-wt/2/pi*0.5);`
- ▶ `fm=-fp;`
- ▶ `dat=[deg' wt' f' fp' fm'];`
- ▶ `save dat dat`
- ▶ `exit` ili `ctrl-d` (`^D`)
- ▶ `gedit dat`
- ▶ `#` znači komentar
- ▶ zapamtiti šta je u kojoj koloni, 1, 2, 3, 4 i 5

gnuplot, file data plotting

- ▶ `plot "dat" using 1:3`
- ▶ `plot "dat" using 1:3 with lines`
- ▶ `plot "dat" using 2:3 with lines`
- ▶ `plot "dat" using 1:3 with lines, "dat" using 1:4 with lines, "dat" using 1:5 with lines`
- ▶ `plot "dat" using 1:3 with lines linewidth 2, \`
`"dat" using 1:4 with lines linetype 2, \`
`"dat" using 1:5 with lines linetype 2`
- ▶ `plot "dat" using 1:3 with lines linewidth 2 title "Vc", \`
`"dat" using 1:4 with lines linetype 2 title "anv 1", \`
`"dat" using 1:5 with lines linetype 2 title "anv 2"`
- ▶ `set xrange [0:360*4]`
- ▶ `set xtics 180; replot`
- ▶ postaje zamorno; vreme je za scripting

gnuplot, prvi script

- ▶ da napravimo script, `gedit slika.gp`
- ▶ ne mora ekstenzija `.gp`, moze bilo šta
- ▶ kucate:

```
set xrange [0:360*4]
set xtics 180
set xlabel "wt [deg]"
set ylabel "Vc"
plot "dat" using 1:3 with lines linewidth 2 title "Vc", \
      "dat" using 1:4 with lines linetype 2 title "anv 1", \
      "dat" using 1:5 with lines linetype 2 title "anv 2"
```

- ▶ `gnuplot`
- ▶ `load "slika.gp"`

gnuplot, još terminala

- ▶ `set terminal png`
- ▶ `set output "slika3.png"`
- ▶ `replot`
- ▶ `set terminal pdfcairo`
- ▶ `set output "slika3.pdf"`
- ▶ `replot`
- ▶ `set terminal jpeg`
- ▶ `set output "slika3.jpg"`
- ▶ `replot`
- ▶ `set terminal wxt`
- ▶ `set output`
- ▶ pogledajte `slika3.png`, `slika3.pdf` i `slika3.jpg`; razlike?

gnuplot, latex terminal

- ▶ napravite script `s14.gp`

```
set terminal latex
set output "s14.tex"
set xrange [0:360*4]
set xtics 180
set xlabel "$\omega t \ ; [\^{\circ}]$"
set ylabel "$v_C$"
plot "dat" using 1:3 with lines linewidth 2 title "$v_C$", \
"dat" using 1:4 with lines linetype 2 title "anv$_1$", \
"dat" using 1:5 with lines linetype 2 title "anv$_2$"
```

- ▶ mora `\` umesto `\` pošto je `\` gnuplot znak za novi red
- ▶ može `load "s14.gp"`
- ▶ a može i sa komandne linije: `gnuplot s14.gp`
- ▶ dobili ste `s14.tex` koji nije standalone

gnuplot, latex terminal, cover

- ▶ napravite tex cover `slika4.tex`:

```
\documentclass{minimal}
```

```
\begin{document}
```

```
\input{s14}
```

```
\end{document}
```

- ▶ može Texmaker, a može i komandna linija `pdflatex slika4`, direktno u pdf
- ▶ može i indirektno, dvi-ps-pdf: `latex slika4, dvips slika4, ps2pdf slika4.ps`
- ▶ može da se „kropuje“, `ps2eps -l slika4.ps, epstopdf slika4.eps`
- ▶ može i `latex slika4, dvipdf slika4`, nema kropovanja

gnuplot, epslatex terminal

- ▶ napravite script

```
set terminal epslatex standalone
set output "slika5.tex"
set xrange [0:360*4]
set xtics 180
set xlabel "$\omega t \ ; \ [^\circ]$"
set ylabel "$v_C$"
plot "dat" using 1:3 with lines linewidth 2 title "$v_C$", \
"dat" using 1:4 with lines linetype 2 title "anv$_1$", \
"dat" using 1:5 with lines linetype 2 title "anv$_2$"
```

- ▶ procesirate kroz gnuplot, dobili ste standalone tex
- ▶ latex `slika5`, ne može `pdflatex slika5` !!!
- ▶ put do pdf na više načina

gnuplot, epslatex color terminal

- ▶ napravite script

```
set terminal epslatex standalone color
set output "slika6.tex"
set xrange [0:360*4]
set xtics 180
set xlabel "$\omega t \ ; \ [^\circ]$"
set ylabel "$v_C$"
plot "dat" using 1:3 with lines linewidth 2 title "$v_C$", \
"dat" using 1:4 with lines linetype 2 title "anv$_1$", \
"dat" using 1:5 with lines linetype 2 title "anv$_2$"
```

- ▶ procesirate kroz gnuplot, dobili ste standalone tex
- ▶ procesirate kroz latex, procesirate preko dvi i ps do pdf
- ▶ pogledate i uporedite sve slike od sada

gnuplot, parametric plot

- ▶ `set parametric`
- ▶ `plot [0:2*pi] cos(t), sin(t), 2*cos(t/2), 2*sin(t/2)`
- ▶ `set xrange [-3:3]`
- ▶ `set yrange [-3:3]`
- ▶ `replot`
- ▶ `set size square`
- ▶ `replot`
- ▶ `set key above box; replot`

gnuplot, another parametric plot

create mystery-curve.gp as follows:

```
set terminal epslatex standalone color size 6, 6
set output "mystery-curve.tex"
```

```
set parametric
set samples 10000
set size square
```

```
set key off
unset xtics
unset ytics
unset border
```

```
plot cos(t)+cos(6*t)/2+sin(14*t)/3, \
sin(t)+sin(6*t)/2+cos(14*t)/3 linewidth 4
```

... and a script file ...


```
#!/bin/sh
gnuplot $1.gp
pdflatex $1.tex
rm -f $1.tex
rm -f $1-inc*
rm -f $1.aux
rm -f $1.log
```

save it as gpmake.sh, just run it,

```
sh gpmake.sh mystery-curve
```

and look for the output

... and the result is ...



gnuplot, polar plot

- ▶ `reset`
- ▶ `set polar`
- ▶ `plot [0:2*pi] t`
- ▶ `plot [0:10*2*pi] t`
- ▶ `set samples 1000`
- ▶ `set size square`
- ▶ `set xrange [-70:70]`
- ▶ `set yrange [-70:70]`
- ▶ `unset key; replot`
- ▶ `plot [0:10*2*pi] t, -t`

gnuplot, logscale, kompleksni brojevi

- ▶ resetujte gnuplot settings, restart to sigurno radi
- ▶ smatraćemo da je $x = \frac{\omega}{\omega_P}$
- ▶ `j={0,1}`
- ▶ `print j**2`
- ▶ `print j*j`
- ▶ `set xrange [0.001:1000]`
- ▶ `set logscale x`
- ▶ `plot 20*log10(abs(1/(1+j*x)))`
- ▶ `set yrange [-80:20]; replot`
- ▶ `plot 20*log10(abs(1/(1+(j*x)**2)))`
- ▶ `set yrange [-160:160]; replot`
- ▶ `set ytics 20; replot`
- ▶ `set samples 1000; replot`
- ▶ `set samples 10000; replot`

gnuplot, fazna karakteristika

- ▶ `plot 180/pi*arg(1/(1+(j*x)**2))`
- ▶ `set yrange [-270:90]; replot`
- ▶ `set ytics 90; replot`
- ▶ `plot 180/pi*arg(1/(1+j*x))`
- ▶ `set yrange [-135:45]; replot`
- ▶ `set ytics 45; replot`

gnuplot, amplitudska i fazna karakteristika, script

napravite skript af.gp:

```
set terminal wxt 0 persist
j={0,1}
set logscale x
set xrange [0.001:1000]
set yrange [-120:40]
set ytics 20 nomirror out
set y2range [-270:90]
set y2tics 90 nomirror out
set samples 1000
set xlabel "log(w/(1 rad/s))"
set ylabel "a [dB]"
set y2label "phi [deg]"
plot 20*log10(abs(1/(1+(j*x)**2))) axes x1y1, \
180/pi*arg(1/(1+(j*x)**2)) axes x1y2
```

gnuplot af.gp

gnuplot, styles and size

```
set xrange [-pi:pi]
set xtics pi
set yrange [-1.5:1.5]
set ytics 1
set style line 1 linetype 1 linewidth 4 linecolor rgb "red"
set style line 2 linetype 2 linewidth 2 linecolor rgb "blue"
plot sin(x) linestyle 1, cos(x) linestyle 2
set terminal postscript eps color size 12cm, 8cm
set output "size1.ps"
replot
set terminal postscript eps color size 4cm, 4cm
set output "size2.ps"
replot
```

gnuplot, margins

```
show margin
set terminal postscript eps color size 12cm, 8cm
set output "margin1.ps"
replot
set lmargin at screen 0.2
set rmargin at screen 0.9
set bmargin at screen 0.2
set tmargin at screen 0.9
show margin
set output "margin2.ps"
replot
set lmargin
show margin
```

gnuplot, axis, border, tics

```
reset  
set terminal wxt 0  
set output  
plot sin(x)  
unset border; replot  
set zeroaxis; replot  
set xtics axis; replot  
set ytics axis; replot  
set zeroaxis linetype -1; replot
```


gnuplot, test, terminal specific parameters

- ▶ vrlo korisna komanda, `test`
- ▶ prikazuje “terminal specific” parametre
- ▶ zaključak: odaberete nekoliko terminala koji vas interesuju (`epslatex`, `png`, `jpeg`, `postscript`, `pdfcairo`), naviknete se na značenje pojedinih parametara
- ▶ kroz `test` se odlično vidi razlika između vektorske i rasterske grafike

gnuplot, test, važniji terminali

```
set terminal png
set output "test.png"
test

set terminal postscript eps
set output "test.eps"
test

set terminal pdfcairo
set output "test.pdf"
test

set terminal jpeg
set output "test.jpg"
test

set terminal wxt 0
set output
test
```

gnuplot, size, ratio

```
reset
set terminal wxt 0
set output
set parametric
plot [0:2*pi] cos(t), sin(t)
set xrange [-2:2]
set yrange [-1:1]
replot
set size square
replot
set size ratio 1
replot
set size ratio 0.5
replot
```

gnuplot, wxt interactive commands

```
reset  
plot exp(x)  
l  
l  
m  
m
```

gnuplot, 3D plot, sampling

```
reset
set xrange [-2*pi:2*pi]
set yrange [-2*pi:2*pi]
splot cos(x)*sin(y)
show isosamples
set isosamples 2; replot
show samples
set samples 10; replot
set samples 100; replot
set isosamples 3; replot
set isosamples 5; replot
set isosamples 7; replot
```

gnuplot, 3D plot, hidden3d

```
show hidden3d
set hidden3d
show hidden3d
replot
set isosamples 50
replot
```

gnuplot, 3D plot, contours

```
show contour  
set contour  
show contour  
replot  
help contour  
set contour both  
replot
```

gnuplot, 3D plot, surface

```
set contour surface  
replot  
unset surface  
replot
```


gnuplot, 3D plot, more contours

```
splot sin(sqrt(x**2+y**2))/sqrt(x**2+y**2) title "sombbrero"  
help cntrparam  
set cntrparam levels discrete 0,0.25,0.5,0.75  
replot  
set surface  
replot  
set contour both  
replot
```

gnuplot, 3D plot, cubic spline

```
set isosamples 10
replot
set cntrparam cubicspline
replot
show view
set view map
replot
unset surface
replot
set size ratio 1
replot
set isosamples 50
replot
```

gnuplot, 3D plot, views

```
show view  
set view  
show view  
replot
```

gnuplot, 3D plot, bojenje

```
set pm3d
replot
unset contour
replot
set view map
replot
set view
replot
set pm3d at b
replot
set pm3d at t
replot
```

gnuplot, 3D plot, 3D parametarski plot

```
reset  
set parametric  
splot [0:20*pi] cos(u), sin(u), u
```

```
set samples 1000  
replot
```

octave, data files

octave

```
z = rand(21);  
save z z  
x = linspace(0, 4 * pi, 361);  
x0 = cos(x);  
y0 = sin(x);  
z0 = exp(x / pi);  
xx = [x0' y0' z0'];  
save xx xx  
ctrl-d
```

gnuplot, data matrix plot

```
splot "z" matrix  
splot "z" matrix with lines  
set hidden3d; replot
```

gnuplot, contours

```
set contour; replot
unset surface; replot
set view map; replot
set size ratio 1; replot
unset key; replot
unset tics; replot
set cntrparam cubicspline; replot
```


gnuplot, palette

```
set pm3d; replot
unset colorbox; replot
unset contour; replot
set palette gray; replot
```

gnuplot, 3D data line

```
reset
```

```
splot "xx" using 1:2:3
```

```
splot "xx" using 1:2:3 with lines title "spirala"
```

gnuplot, 3D plot, frekvencijski primer

```
j={0,1}
set xlabel "sigma"
set ylabel "j omega"
set xrange [-3:3]
set yrange [-3:3]
set zrange [0:5]
splot abs(1/(1+(x+j*y)**2))
```

gnuplot, 3D plot, frekvencijski primer, contours

```
set contour; replot
set contour surface; replot
set contour both; replot
set cntrparam levels discrete 1,2,3; replot
unset surface; replot
set view map; replot
set xrange [-2:2]
set yrange [-2:2]
replot
set cntrparam cubicspline; replot
set size square; replot
```

gnuplot, frekvencijski primer, colorplot

```
set view; replot  
set pm3d; replot  
set pm3d at t; replot  
set pm3d at st; replot  
set pm3d at b; replot  
set pm3d map; replot
```

gnuplot, set view equal

```
set xrange [0: pi]
set yrange [0:1]

set contour both
unset surface
set cntrparam levels discrete 0.2, 0.4, 0.6, 0.8, 1, 1.2, 1.4
set pm3d

set isosamples 41


set xlabel "x"
set xtics ("0" 0, "pi" pi)
set ylabel "y"
set ytics ("0" 0, "1" 1)

set label at pi/2, 0.5, 2 "|sin(z)| = |sin(x + i y)|" center

set view equal

splot sqrt(sin(x)**2*cosh(y)**2+cos(x)**2*sinh(y)**2) notitle
```

gnuplot, set view equal, fig.gp, extended



gnuplot, planeta, 3D

```
set terminal wxt size 600,600 persist

unset key; unset border; unset tics

set lmargin screen 0.05; set bmargin screen 0.05
set rmargin screen 0.95; set tmargin screen 0.95

set mapping spherical
set angles degrees
set hidden3d
set xyplane at -1
set view 60, 120

set parametric
set isosamples 37
set urange[0:360]; set vrange[-90:90]
set size square

plot cos(v)*cos(u), cos(v)*sin(u), sin(v) w l lc rgb "cyan", \
      "world_110m.txt" with lines lc rgb "blue" lw 2
```


gnuplot, planeta, 3D



gnuplot, planeta, 2D

```
set terminal pdfcairo size 360 * 0.01, 180 * 0.01
set output "map.pdf"
```

```
unset key
unset border
unset tics
```

```
set style line 1 lc rgb "gray" lt 1 lw 1
set style line 2 lc rgb "black" lt 1 lw 1
```

```
set lmargin 0
set rmargin 1
set tmargin 1
set bmargin 0
```


```
set xrange [-180:180]
set yrange [-90:90]
```

```
plot "world_110m.txt" w filledcu ls 1, \
      "world_110m.txt" w l ls 2
```

gnuplot, planeta, 2D



gnuplot, zaključak



preporuka: <http://gnuplot.sourceforge.net/demo/>